
ANEXA VI

PRIMUL PROGRAM ÎN LIMBAJ DE ASAMBLARE

I.1. Enunţul problemei


Se dau 5 numere a, b, c, d şi e cu valorile:

a = 4

b = 2

c = 6

d = 3

e = 9

Să se calculeze expresia:

 x = a * b - c / d + e.


Operaţiile se efectuează în mulţimea numerelor întregi.

I.2. Alegerea tipului de date


Ca în orice problemă rezolvată cu ajutorul calculatorului, această etapă este deosebit de importantă, de ea depinzând acurateţea rezultatelor obţinute. Pentru numere întregi, alegerea se face astfel:

· dacă numerele aparţin intervalului [-128; +127] sau [0; 255], se alege memorarea pe 1 byte , folosindu-se descriptorul db (define byte).

· dacă numerele aparţin intervalului [-32768; +32767] sau [0; 65535], se alege memorarea pe 2 bytes (1 cuvânt) , folosindu-se descriptorul dw (define word).

· dacă numerele aparţin intervalului [-231-1; +231-1] sau [0;  232-1], se alege memorarea pe 4 bytes (2 cuvinte), folosindu-se descriptorul dd (define double [word]).

· dacă numerele aparţin intervalului [-263-1; +263-1] sau [0;  264-1], se alege memorarea pe 8 bytes (4 cuvinte), folosindu-se descriptorul dq (define quad [word]).


În mod normal, operaţiile se efectuează pe maxim 16 biţi (1 cuvânt). Pentru lucrul cu valori mai mari, se folosesc registrele extinse, pe 32 biţi.


În acest caz, pentru că valorile sunt mici, se alege memorarea pe 1 byte, folosind descriptorul db.

I.3. Schema logică a programului


Pentru a implementa corect prioritatea operatorilor aritmetici implicaţi, se folosesc două variabile temporare t1 şi t2, care memorează rezultatele operaţiilor de înmulţire (a * b), respectiv,  de împărţire (c / d).


Schema logică este cea din figura I.1.:

Figura I.1: Schema logică a programului.

I.4. Scrierea programului


Într-un limbaj de asamblare, fiecare instrucţiune este scrisă pe un rând separat.

Directivele se scriu precedate de "." (Exemplu .stack).

Comentariile se scriu pe linii separate sau după instrucţiuni, precedate fiind de ";" (Exemplu: ;definire variabile) şi se folosesc pentru a înlesni înţelegerea programului.

Etichetele se scriu urmate de ":" (Exemplu: repeta:) şi se folosesc pentru a defini destinaţiile salturilor sau pentru a delimita blocurile de program.


Orice program scris în limbaj de asamblare are atât instrucţiunile, cât şi datele definite în interiorul unor segmente. Aceste segmente sunt organizate conform unui model de memorie. În versiunile mai noi ale asambloarelor au apărut directive simplificate de definire a segmentelor şi modelelor de memorie.


În program se vor folosi două segmente: unul pentru date şi unul pentru cod. Acestea sunt definite cu directivele .data, respectiv .code. Segmentul de date va cuprinde definirea datelor folosite, iar cel de cod instrucţiunile propriu-zise ale programului.

Textul sursă al programului este:

.model small    ;definire model memorie

.data           ;definire segment de date

  a db 4        ;definire variabile

  b db 2

  c db 6

  d db 3

  e db 9

  t1 db ?       ;definire rezultate

  t2 db ?

  x db ?

.code           ;definire segment de cod

 start:         ;eticheta

   mov ax,@data ;initializare segment de date

   mov ds,ax

   mov al,a     ;al=a

   mul b        ;al=al*b

   mov t1,al    ;t1=al

   xor ax,ax    ;ax=0;

   mov al,c     ;al=c

   div d        ;al=al div d

   mov t2,al    ;t2=al

   mov al,t1    ;al=t1

   sub al,t2    ;al=al-t2

   add al,e     ;al=al+e

   mov x,al     ;x=al

   mov ah,4Ch   ;4Ch (cod serviciu)

   int 21h      ;apel intrerupere DOS

 end start      ;sfarsit program
Se analizează acum fiecare grup de instrucţiuni important:

  a db 4        ;definire variabile

  b db 2

  c db 6

  d db 3

  e db 9


Variabilele se definesc cu sintaxa generală nume d* valoare, unde "*" poate fi "b"(byte), "w"(word), "d"(double) sau "q"(quad), corespunzător tipului de dată şi memoriei ocupate.

  t1 db ?       ;definire rezultate

  t2 db ?

  x db ?


Rezultatele intermediare şi finale sunt, la rândul lor, tot variabile. Sintaxa generală este: nume d* ?. Spre deosebire de variabilele iniţializate cu valori, o astfel de definiţie doar rezervă memorie pentru variabilă, aceasta urmând să primească drept valoare rezultatele calculelor.


Instrucţiunea mov se utilizează pentru transferul datelor elementare, şi are forma generală mov destinaţie, sursă. 

   mov ax,@data ;initializare segment de date

   mov ds,ax


Registrul ds conţine adresa segmentului de date (@data), şi trebuie iniţializat la începutul programului. Deoarece acest lucru nu se poate realiza direct, se foloseşte ca intermediar registrul de uz general ax. După această iniţializare, se folosesc datele definite în segmentul de date.

mov al,a     ;al=a

mul b        ;ax=al*b

mov t1,al    ;t1=al
Se calculează termenul t1. Majoritatea operaţiilor aritmetice se fac cu ajutorul unui registru special, operand implicit în unele operaţii (ca de exemplu cea de înmulţire: mul b): acumulatorul. Acesta este al pentru operaţii pe 8 biţi, ax pentru operaţii pe 16 biţi sau eax pentru operaţii pe 8 biţi. Operaţia de înmulţire generează de obicei rezultatul pe un număr dublu de biţi. Aici, înmulţind doi operanzi pe 8 biţi (registrul al şi variabila b), rezultatul va fi pe 16 biţi, memorat implicit în registrul ax. Pentru că am lucrat cu numere mici, putem ignora primii 8 biţi cei mai semnificativi, cuprinşi în registrul de 16 biţi ax, adică registrul de 8 biţi ah. De aceea, în t1 se memorează doar conţinutul registrului al.

xor ax,ax    ;ax=0;

mov al,c     ;al=c

div d        ;al=al div d

mov t2,al    ;t2=al
Operaţia inversă, de împărţire, porneşte de la operandul deîmpărţit memorat pe 16 biţi şi furnizează câtul pe 8 biţi. Pentru a asigura că partea cea mai semnificativă a acumulatorului ax este nulă (altfel se obţin rezultate eronate), se iniţializează ax cu zero. Această operaţie este realizată printr-un sau exclusiv pe biţi (xor) al acumulatorului cu el însuşi. Se încarcă partea cea mai puţin semnificativă a acumulatorului cu valoarea variabilei c. Instrucţiunea div efectuează împărţirea întreagă, memorând în registrul al câtul şi în registrul ah restul acesteia.

mov al,t1    ;al=t1

sub al,t2    ;al=al-t2

add al,e     ;al=al+e

mov x,al     ;x=al


Se efectuează ultimele două operaţii: scăderea şi adunarea, urmate de memorarea rezultatului în variabila x.

   mov ah,4Ch   ;ah=4Ch (cod serviciu)

   int 21h      ;apel intrerupere DOS

Întoarcerea la sistemul de operare se face prin apelul întreruperii DOS, cu codul 21h, primind ca parametru în registrul ah codul 4Ch, ceea ce înseamnă tocmai serviciul de întoarcere la sistem. Se observă că valoarea calculată şi memorată în operandul x a rămas şi în registrul al, acesta fiind folosit de serviciul 4Ch al întreruperii 21h pentru a semnala eventuale erori la execuţie.


Toate instrucţiunile au fost încadrate de eticheta start astfel:

start:         ;eticheta

..

end start      ;sfarsit program
I.5. Crearea fişierului ".asm"

Pentru introducerea textului unui program în limbaj de asamblare se  foloseşte un editor de texte, cu condiţia ca acesta să nu insereze caractere speciale de formatare în cadrul textului.


Se tastează textul linie cu linie, apoi se salvează cu extensia .asm (de exemplu: primul.asm).

I.6. Asamblarea, editarea legăturilor şi execuţia


Odată salvat fişierul, se trece la asamblare. Aceasta se face cu un programul specializat numit asamblor. Există mai multe asemenea programe, cele mai cunoscute fiind tasm (Turbo Assembler, produs de firma Borland) şi masm (Microsoft Assembler, produs de firma Microsoft). Se tastează:

C:\>tasm primul.asm
sau

C:\>masm primul.asm

Dacă textul programului a fost tastat corect, pe ecran va apare (la Turbo Assembler):

C:\>tasm primul.asm

Turbo Assembler  Version 4.1  Copyright (c) 1988, 1996 Borland International

Assembling file:   primul.asm

Error messages:    None

Warning messages:  None

Passes:            1

Remaining memory:  384k

Programul tasm.exe sau masm.exe trebuie să se afle în acelaşi director, sau să fie declarat în variabila de sistem PATH, definită în autoexec.bat.


Rezultatul va fi un fişier cu extensia .obj. Acest fişier conţine codul obiect al programului. Pentru a îl aduce în formă executabilă, trebuie efectuată operaţia de editare de legături (link-editarea), cu ajutorul unui alt program specializat, editorul de legături. Şi aici există două link-editoare mai cunoscute: tlink, respectiv link. Se tastează:

C:\>tlink primul.obj
sau

C:\>link primul.obj

Pe ecran va apare:

C:\>tlink primul.obj

Turbo Link  Version 7.1.30.1. Copyright (c) 1987, 1996 Borland International

Warning: No stack


Se observă un mesaj de avertizare generat datorita faptului că nu s-a folosit un segment de stivă (în acest caz, nu era necesar).


Rezultatul va fi, de această dată, un fişier direct executabil, primul.exe. Acesta poate fi executat tastând:

C:\>primul
I.7. Urmărirea pas cu pas a execuţiei


Pentru a observa modul cum se execută un program, acesta poate fi încărcat într-un program specializat, numit debugger şi executat pas cu pas. Un exemplu de debugger este td (Turbo Debugger, produs de Borland). Acesta oferă o interfaţă multi-fereastră (în mod text), cu ajutorul căreia se pot urmări, printre altele:

· instrucţiunea care va fi executată;

· conţinutul segmentului de date;

· conţinutul registrelor.


Încărcarea programului în debugger pentru execuţie pas cu pas se face tastând (de exemplu):

C:\>td primul.exe


Figura I.2. prezintă interfaţa programului td, în care s-a încărcat programul primul.exe:


Figura I.2. Programul încărcat în Turbo Debugger


Se observă că ecranul este împărţit în mai multe regiuni:

· regiunea din stânga sus conţine codul programului. Se observă că numele variabilelor au fost înlocuite cu numere, care reprezintă deplasamente (offset) în cadrul segmentului de date.

· regiunea din stânga jos prezintă conţinutul segmentului de date. Se observă zona de memorie ce conţine variabilele folosite de program (figura I.3.).

· regiunea din dreapta sus conţine registrele procesorului şi indicatorii de condiţie. Aici se urmăresc efectele instrucţiunilor mov.

· regiunea din dreapta jos prezintă conţinutul segmentului de stivă, în acest caz, nu s-a folosit stiva, deci conţinutul acestei zone este nedefinit.


Figura I.3. Conţinutul segmentului de date


Conţinutul registrului ax la fiecare pas al execuţiei programului, execuţie ce se realizează prin apăsarea tastei F7este prezentat în figura I.4.

Instrucţiunea
Conţinut ax
Observaţii

mov ax,@data
1BB7h
ax=@data

mov ds,ax
1BB7h
ds=ax

mov al,a
1B04h
al=4; ah rămâne 1Bh

mul b
0008h
ax=4*2=8

mov t1,al
0008h
t1=al

xor ax,ax
0000h
ax=0

mov al,c
0006h
al=6; ah rămâne 0

div d
0002h
al=6 div 3=2; ah=6 mod 3=0

mov t2,al
0002h
t2=al

mov al,t1
0008h
al=t1

sub al,t2
0006h
al=al-2=8-2=6

add al,e
000Fh
al=al+9=6+9=15=0Fh

mov x,al
000Fh
x=al

mov ah,4ch
4C0Fh
ah=4Ch; al rămâne 0Fh

int 21h
4C0Fh
ieşire

Figura I.4. Evoluţia registrului ax

Varianta de program prezentată are operanzii iniţializaţi la definire. Spre deosebire de limbajele evoluate de programare unde sunt instrucţiuni sau funcţii de intrare/ieşire, ca de exemplu read()/write() in Pascal, scanf()/printf() în C, limbajele de asamblare nu sunt înzestrate cu astfel de funcţii. Programele apelează la întreruperi pentru realizarea operaţiilor de Intrare/Ieşire pentru şiruri de caractere, urmând a fi scrise proceduri de conversie de la şirurile de caractere la numere.

În continuare se prezintă o variantă a programului în care variabilele a, b, c, d şi e se citesc de la tastatură iar rezultatul, din variabila x, este afişat pe ecran.

Se definesc variabilele  msgA, msgB, msgC, msgD, msgE, msgX, msgEr, iniţializează cu şiruri de caractere care se afişează.

include conv.mac


dosseg


.model small


.stack 100h


.data

msgA
db 'A=$'

msgB
db 'B=$'

msgC
db 'C=$'

msgD
db 'D=$'

msgE
db 'E=$'

msgX
db 'Expresia are valoarea:$'

msgEr
db 'Impartire la 0!$'

A
dw 0

B
dw 0

C      dw 0

D      dw 0

E      dw 0

X
dw 0

t1
dw 0

t2
dw 0

buf
db 7 dup(?) ;buffer in care se citesc numerele

nr
dw ?

baza dw 10 ;baza in care se realizeaza conversia


.code

start:


mov ax,@data


mov ds,ax


;citirea si conversia lui A


afis msgA


citsn buf


atoi buf,A,baza


;citirea si conversia lui B


afis msgB


citsn buf


atoi buf,B,baza


;citirea si conversia lui C


afis msgC


citsn buf


atoi buf,C,baza


;citirea si conversia lui D


afis msgD


citsn buf


atoi buf,D,baza


;citirea si conversia lui E


afis msgE


citsn buf


atoi buf,E,baza


mov ax,A


mul B

;ax=A*B


mov t1,ax     ;t1=ax


xor ax,ax     ;ax=0


mov ax,c      ;ax=c


cmp d,0


je Eroare


div d         ;ax=C/D


mov t2,ax     ;t2=ax


mov ax,t1


sub ax,t2     ;ax=t1-t2


add ax,e


mov x,ax      ;x=ax+e


;afisarea lui X


afis msgX


afisn x,baza


jmp Iesire

Eroare:


afis msgEr

Iesire:


mov ax,4c00h


int 21h


end start
În plus, această versiune a programului testează cazul în care operandul d are valoarea zero şi afişează un mesaj corespunzător, definit în variabila msgEr.

I.8. Erori


Există mai multe tipuri de erori care pot apărea în fazele de rezolvare a unei probleme cu ajutorul unui program scris în limbaj de asamblare. Iată câteva dintre principalele tipuri de erori:

· erori de compilare (de exemplu, se tastează ad în loc de add). De cele mai multe ori, acestea sunt depistate în timpul procesului de asamblare (când duc la erori de sintaxă, neacceptate de asamblor), dar pot trece de acesta (de exemplu, dacă se tastează add în loc de adc, asamblorul nu va semnala o eroare, dar probabil rezultatul operaţiei va fi eronat).

· erori la link-editare sunt acele erori care se propagă în codul obiect după asamblare, dar sunt semnalate de link-editor. De obicei, acestea se datorează referirii incorecte a datelor aflate în segmente diferite sau folosirii unui model de memorie inadecvat.

· erori de execuţie sunt acele erori care nu sunt semnalate de asamblor, dar determină neobţinerea rezultatelor aşteptate, sau chiar obţinerea unor erori de sistem. De exemplu, o împărţire prin zero va genera terminarea anormală a execuţiei şi afişarea unui mesaj: "Your program caused a divide overflow error. If the problem persists, contact your program vendor." ). Corectarea acestora se face prin urmărirea pas cu pas a execuţiei (debugging).

I.9. Concluzii


Acest exemplu de program simplu a ilustrat unele din cele mai folosite concepte în programarea în limbaj de asamblare, precum: definirea datelor, directivele, regiştrii, operaţiile aritmetice, asamblarea, link-editarea, execuţia pas cu pas şi depanarea.

START





t1 = a * b





t2 = c / d





x = t1 - t2





x = x + e





STOP








